
Простейшая программа на Паскале

	Program HelloWorld;
	в Паскале все программы должны начинаться с объявления имени программы

	BEGIN
	

	Writeln('Hello!')
	это подпрограмма, которая печатает на экране текстовые и численные значения.

	END.
	

	
	

	BEGIN ... END.
	между этими ключевыми словами размещаются инструкции алгоритма, или операторы программы.

	'Hello!'
	так в Паскале кодируются текстовые строки.


Текстовые строки

Текст строки заключается в одинарные кавычки.  Если сам текст должен содержать одинарные кавычки, то каждая одинарная кавычка в тексте заменяются на две подряд идущие кавычки.

Пример строки:

	I can't read it !
	Нам нужна такая текстовая строка

	'I can''t read it !'
	А так её надо оформить в Паскаль-программе


Несколько операторов

	Program a_2x2;
	во всех языках программирования (в т.ч. в Паскале) есть свои правила написания имён.

	BEGIN
	

	Writeln('2+2=', 2+2);
	если между словами BEGIN и END размещены несколько операторов, то операторы надо отделять друг от друга точкой с запятой ";"

	Writeln('2x2=', 2*2)
	После этого оператора точку с запятой можно не ставить

	END.
	


Процедуры вывода значений на экран – Writeln и Write
	Writeln(значение1, значение2, …)
	Последовательно печатает на экране значения всех выражений, текстовых и числовых. По окончании распечатывания списка переводит курсор на новую строку.

	Write(значение1, значение2, …)
	Последовательно печатает на экране значения всех выражений, текстовых и числовых. По окончании распечатывания списка курсор остаётся на текущей строке.

	Writeln(314.15926535);
	Печатает вещественное число в формате "с плавающей запятой":


3.1415926535Е+0002

	Writeln(314.15926535:12:8);
	Печатает вещественное число в формате "с фиксированной запятой", последнее число (после двоеточия) указывает количество цифр в дробной части:


314.15926535

	Writeln(314.15926535:6:2);
	Если количество цифр "после запятой" в формате меньше, чем у числа, то происходит округление последней цифры:


314.16


Типы чисел в Паскале и арифметические операции.

Паскаль поддерживает целые (integer) и вещественные числа (real). Для чисел поддерживаются 4 основные арифметические операции, однако для для целых и вещественных чисел операции выполняются по-разному: в программировании важен тип данных, с которыми работает программа.

	:=
	Присвоение значения
(не арифметическая операция в точном смысле слова, применима к любым типам)

	+
	Сложение 
 (для любых чисел)

	-
	Вычитание 
 (для любых чисел)

	*
	Умножение 
 (для любых чисел)

	/
	Деление вещественное 
 (для любых чисел, результат всегда вещественный)

	div
	Деление целочисленное 
 (только для целых чисел !)

	mod
	Взятие остатка от деления 
 (только для целых чисел !)


Если в арифметической операции хотя бы один операнд вещественный, то и результат операции будет вещественный. Если к двум целым числам применяется операция вещественного деления, то результат также будет вещественный.

Примеры:

	(1+2.5)*5.8/0.1
	3.14159265-(2.71828*2.71828)
	333 / 45

	1234 mod 1000
	((598932 div 951) + 21)*15 mod 4
	34 * 45 div 13 mod 4


Переменные

Во всех языках программирования (в т.ч. и в  Паскале) для хранения значений используются переменные. По правилам Паскаля, каждая переменная должна иметь уникальное имя, а также особую характеристику – тип. Определение имени и типа переменной называется объявлением переменной. В Паскале переменные объявляются в специальных секциях объявления переменных.

Имена переменных

Каждая переменная должна иметь уникальное имя (в пределах функции, процедуры или программы). По имени переменные идентифицируются. В Паскале есть правила написания имён переменных:

· Первый символ имени переменной – буква английского алфавита (A..Z) или символ подчерка ( _ ).

· Последующие символы имени переменной – буква английского алфавита, символ подчерка или цифра (0..9).

Заглавные и строчные буквы считаются эквивалентными,
поэтому имена переменных 'ABCD', 'abcd', 'Abcd', 'aBCd' или 'abcD' являются эквивалентными.

Примеры:

	Правильные имена
	a,b,i, z, Discriminant, zoO, RAT, _1st_floor, root,  ___

	Неправильные имена
	3rd_number, 1234, @ghfjg, *aster, rock-n-roll, plug'n'play


Типы переменных
Тип переменной определяет, какие операции можно выполнять с переменной, и – каким будет результат операции.

В Паскале есть несколько стандартных типов переменных, а также по определённым правилам можно создавать свои типы данных. Ниже приводятся некоторые из стандартных типов Паскаля:

Целые числа:

	integer
	Знаковое целое число,
размер 4 байта,
диапазон значений –2'147'483'648 .. 2'147'483'647

	longint
	Знаковое целое число,
размер 4 байта,
диапазон значений –2'147'483'648 .. 2'147'483'647

	shortint 
	Знаковое целое число,
размер 1 байт,
диапазон значений –128 .. 127

	int64
	Знаковое целое число,
размер 8 байтов, диапазон –9'223'372'036'854'775'808 .. 9'223'372'036'854'775'807

	byte
	Беззнаковое целое число,
размер 1 байт,
диапазон значений 0 .. 255

	word
	Беззнаковое целое число,
размер 2 байта,
диапазон значений 0 .. 65'535

	cardinal
	Беззнаковое целое число,
размер 4 байта,
диапазон значений 0 .. 4'294'967'295


Вещественные числа:

	single
	Размер 4 байта,
диапазон   ±1.5·10-45 .. ±3.4·1038,
значащих цифр – 7-8

	double
	Размер 8 байтов,
диапазон  ±5.0·10-324 .. ±1.7·10308,
значащих цифр – 15-16

	extended
	Размер 10 байтов,
диапазон ±3.6·10-4951 .. ±1.1·104932,
значащих цифр – 19-20

	real
	Размер 8 байтов,
диапазон  ±5.0·10-324 .. ±1.7·10308,
значащих цифр – 15-16

	real48
	Размер 6 байтов,
диапазон   ±2.9·10-39 .. ±1.7·1038,
значащих цифр – 11-12


Символьные и строковые данные

	char
	Т.н. ASCII-символ, размер 1 байт.
Задание значений: 'A', 'a', '0', '*', #48, #$33, '''', '"'

	string
	Строка ASCII-символов, длина строки практически неограничена (до 2 млрд. символов).
Задание значений: 'Abracadabra', 'it''s my life', '0'#48#$33''''

	shortstring
	Строка ASCII-символов, максимальная длина строки 255 символов.

	widechar
	Т.н. Unicode-символ, размер 2 байта.

	widestring
	Строка Unicode-символов, длина строки практически неограничена (до 1 млрд. символов).


 Объявления переменных

В Паскале переменные объявляются в специальных секциях объявления переменных. Секции объявления переменных должна быть расположена перед исполняемыми операторами, т.е. до операторов BEGIN .. END. 
Секция объявления переменных начинается ключевым словом VAR. Структура секции переменных:

Var

   <имя-1>,<имя-2>,...<имя-n>: <тип данных-1>;

   <имя-x>,<имя-y>,...<имя-z>: <тип данных-2>;

      ... ...

Пример:

Var

   i: integer;   a, b, c, d: real;

   name, city: string;
Структура программы с переменными:

// === Example 1 ===

Program Average_Harmonical;

Var

   a, b, c: extended;

BEGIN

   Write('Please enter numbers A and B: ');

   Readln(a,b);

   c := 1/((1/a + 1/b)/2);                     // c:=2*a*b/(a+b);

   Writeln('Average Harmonical of A and B: ',c:0:20);

   Readln;

END.

// === Example 2 ===
Program Average_Squadric;

Var

   a, b, c: extended;

BEGIN

   Write('Please enter numbers A and B: ');

   Readln(a,b);

   c := SQRT ( (a*a + b*b) /2);

   Writeln('Average Squadric of A and B: ',c:0:20);

   Readln;

END.

Использование функций

Функции – это подпрограммы, которые можно вызывать по имени. Функции получают на вход некоторое количество параметров, или аргументов, и в результате работы возвращают одно значение. Аналогия с математическими функциями! В Паскале есть стандартные фукции (уже написанные, и которые можно использовать), а также есть возможность написания собственных функций. Функции различаются по имени, имя функции – идентификатор.

Некоторые стандартные функции Паскаля/Дельфи

	abs
	Возвращает модуль числа (целого или вещественного). Тип результата совпадает с типом аргумента

	sqrt
	Возвращает значение кв.корня из неотрицательного числа. Результат – всегда вещественное число.

	sin
	Вычисляет синус угла. Аргумент задаётся в радианах, результат – всегда вещественное число

	cos
	Вычисляет косинус угла. Аргумент задаётся в радианах, результат – всегда вещественное число

	arctan
	Вычисляет арктангенс угла. Результат – всегда вещественное число

	exp
	Вычисляет ex, где x – аргумент. Результат – всегда вещественное число

	ln
	Вычисляет натуральный логарифм числа.
Аргумент – любое положительное число, Результат – всегда вещественное число

	trunc
	Отбрасывает дробную часть числа. Результат – всегда целое число

	round
	Округляет число до ближайшего целого. Результат – всегда целое число

	int
	Отбрасывает дробную часть числа. Результат – всегда вещественное число

	frac
	Отбрасывает целую часть числа. Результат – всегда вещественное число, знак совпадает со знаком аргумента.

	chr
	Возвращает ASCII‑символ по его номеру. Аргумент – целое число, результат – ASCII‑символ.

	ord
	Возвращает номер ASCII‑символа. Аргумент – ASCII‑символ, результат – целое число

	length
	Возвращает длину строки в символах. Аргумент – ASCII‑ или Unicode‑строка, результат – целое число

	pos
	Возвращает позицию подстроки в строке.
Аргументы – (1) искомая подстрока, (2) исходная строка.
Результат – положительное целое число, либо ноль (если подстрока отсутствует в строке).

	copy
	Возвращает подстроку из строки. 
Аргументы – (1) исходная строка, (2) первый символ подстроки, (3) длина вырезаемой подстроки.
Результат – новая строка (строковое значение, которое можно присвоить строковой переменной).


Функциям нельзя присваивать значения. В арифметических, логических и иных выражениях функции могут входить только в правую часть (в которой описывается, как вычислить значение выражения). Алгоритм вычисления значения функции (называется телом функции) должен описываться в программе специальным образом.

Набор параметров функции называется списком аргументов, они различаются только по порядку их перечисления при вызове функции. Тип передаваемых в функцию значений обязан совпадать либо быть совместимым по присваиванию с типами аргументов функции.

Примеры использования функций:
Program Compute_ArcSin;

   //  Эта программа вычисляет значение Arcsin(x)

   //  Однако в ней нет ни проверки, ни защиты от ввода некорректных значений X
   //  (как известно, (y ( x=sin(y) ( [-1,1]) 

Var

   x, Arcsin_X: extended;

BEGIN

   Write('Please enter X: ');

   Readln(x);

   Arcsin_X:= arctan( x / (1-sqr(x)) );

   Writeln('Arcsin(X) = ', Arcsin_X:0:20);

   Readln;

END.
Некоторые стандартные процедуры Паскаля/Дельфи

	delete
	Удаляет часть строки. 
Аргументы – (1) исходная строка, (2) первый удаляемый символ, (3) количество удаляемых символов.

	str
	Записывает текстовое представление значения числовой переменной в строковую переменную, формат значения.

Аргументы – (1) числовая переменная и формат, (2) строковая переменная для результата.

	val
	Пытается распознать значение числа (целого или вещественного), записанного в строке.

Аргументы – (1) исходная строка, (2) числовая переменная для результата, (3) переменная для кода ошибки.

	write,
writeln
	Записывает в текстовой форме значения (числовые, строковые, символьные и логические) - на консоль (экран) или в текстовый файл. writeln добавляет перевод строки после вывода всех значений. 

	read,
readln
	Считывает из консоли или текст.файла значения переменных (числовых, строковых и символьных типов). При вводе с клавиатуры, если значений не хватает, то процедуры ожидают ввода всех требуемых значений. readln, после ввода всех значений, пропускает весь текст до перевода строки, или конца файла – для последняя строка. 


Операторы алгоритмических структур

Операторы алгоритмических структур позволяют изменить порядок выполнения операторов в зависимости от некоторых условий, значение которых проверяется в ходе выполнения программы. Варианты: ветвление в зависимости от значения условия, циклическое выполнение в зависимости от значения условия.

Оператор условного перехода ( IF )

Оператор условного перехода позволяет выбрать одно из двух ветвлений алгоритма в зависимости от значения логического выражения. Синтаксис оператора:


<if-оператор> ::= if <лог.выр-е> then <оператор1> [ else <оператор2> ]

Если <лог.выр-е> истинно – выполняется <оператор1>; в противном случае: если описана ветка ELSE выполняется <оператор2>. Далее выполняются следующие операторы. ВАЖНО: оператор IF не позволяет "вернуться" к предыдущим операторам. ВАЖНО: в Паскале/Дельфи в ветках THEN и ELSE может быть только один оператор!

Пример использования оператора IF:

   //  фрагмент решения кв.уравнения

D := B*B – 4*A*C;

if D < 0

  then Writeln('Нет корней!')

  else Writeln('X1=', (-B-sqrt(D))/2/A, ' , X2=', (-B+sqrt(D))/2/A);

Операции сравнения

	<
	меньше
	Операции применимы к:

· целым и вещественным числам

· логическим значениям

· символам и строкам

· перечислениям (*)

· указателям (*)
	Операции неприменимы к:

· массивам и записям

· файлам

	>
	больше
	
	

	=
	равно
	
	

	<=
	меньше или равно
	
	

	>=
	больше или равно
	
	

	<>
	не равно
	
	


Операции сравнения имеют самый низкий приоритет (это влияет на правила расстановки скобок).

Логические операции

	OR
	дизъюнкция (логич.сложение)
	Для этих операций есть одноимённые побитовые операции над целыми числами! Требуется использовать скобки при написании сложных логических выражений!

	AND
	конъюнкция (логич.умножение)
	

	NOT
	отрицание
	

	XOR
	исключающее ИЛИ (эквивалент "не равно")
	

	=
	"равно" (эквивалентность)
	Особенностей нет.

	<>
	"не равно" (эквивалент XOR)
	

	<=
	эквивалент импликации (!)
	


Примеры логических выражений с операциями сравнения и логическими операциями

N_is_a_digit := (N >= 0) and (N < 10);

X_is_large   := (X < -3.5E+10) or (X > 5.5E11);

delta    := 1E-10;                       // эквивалент (=10-10 

NearAorB := (abs(A-x) < delta) or (abs(A-x) < delta);

Операторные скобки ( BEGIN..END )
Группа операторов, заключённых в операторные скобки, интерпретируются как один "составной" оператор. Такой приём активно используется в операторах условного перехода и всех операторах циклов. Синтаксис:


begin [ <оператор> {[ ; <оператор> ]} ] end
Пример использования операторных скобок

   //  фрагмент решения кв.уравнения с разбором значения дискриминанта

D := B*B – 4*A*C;

if D >= 0

  then begin
    b1 := -0.5*B/A;                   // эквивалент (–B/2A)

    if D=0                            // это вложенный оператор IF
      then Writeln('X1=X2=', b1)      // при D=0 печатаем 1 значение

      else begin                      // иначе (D>0) – печатаем 2 значения

        d1 := sqrt(D);

        Writeln('X1=', b1-d1, ' , X2=', b1+d1);

        end
  else Writeln('Нет корней!');        // иначе (D<0) – нет корней!

Оператор безусловного перехода ( GOTO )

Иногда требуется переход к произвольному оператору. Это можно выполнить с помощью оператора безусловного перехода GOTO. Оператор GOTO может выполнять переход только к меткам – операторам, помеченных специальными идентификаторами.

Правила использования оператора GOTO:

1) создать блок объявления меток LABEL

2) Перечислить имена меток, которые будут использованы в программе

3) В нужных местах алгоритма (куда надо делать не‑последовательные переходы) поставить метки: <имя метки>:

4) В нужных местах алгоритма (откуда надо делать не‑последовательные переходы) поставить операторы goto <имя метки>
Пример использования IF и GOTO:

	Program ComputeFactorial;
   //  вычисление факториала N!

Var

  N,i,Factorial: integer;

Label

  DoLoop;

BEGIN

  Readln(N);

  Factorial := 1;

  i := 2;

DoLoop:

  if i<=N

    then begin
      Factorial := Factorial*i;

      i:=i+1;

      goto DoLoop
      end;

  Writeln('N!=', Factorial);

END.

Синтаксическая диаграмма оператора IF:

Блок-схемы, экивалентные оператору условного перехода

В программе:

	   // проверяем, что значение X есть корень уравнения

if A*x*x + B*x + C = 0

then
  Writeln (X, ' is the root of the equation')

else
  Writeln (X, ' is not a root of the equation')


	   // устанавливаем A = max {A, B}

if A < B

then
  A := B




Синтаксическая диаграмма блока операторов в операторных скобках:

Пример:

if A>B
// код, упорядочивающий значения A ≤ B
then begin
   t:=A;

   A:=B;

   B:=t

   end
Операторы циклов

Циклы – повторяемое несколько раз подряд выполнение каких-либо действий. В Pascal есть три вида циклов: 

· цикл FOR
– повторяет действия заданное количество раз; 

· цикл WHILE
– повторяет действия до тех пор, если/пока выполняется заданное условие;

· цикл REPEAT-UNTIL
– прекращает повтор действий, если заданное условие выполнилось. 

Оператор цикла с предусловием ( WHILE )

Цикл WHILE (цикл с предусловием): каждая итерация начинается с проверки условия продолжения цикла – если оно истинно, выполняются действия, затем всё повторяется… Если условие ложно перед первой итерацией, то цикл не выполняется ни разу.


<while-оператор> ::= while <лог.выр-е> do <оператор>
Здесь <лог.выр-е> называется условием цикла, а <оператор> – телом цикла.

Оператор цикла с постусловием ( REPEAT..UNTIL )

Цикл REPEAT-UNTIL (цикл с постусловием): на каждой итерации сперва выполняются действия, после чего проверяется условие выхода из цикла – если оно истинно, то цикл прерывается, а иначе – всё повторяется… Особенность цикла REPEAT-UNTIL – его тело выполняется хотя бы один раз при любом условии.


<repeat-оператор> ::= repeat [<опер1>] [{; <операторi>}] until <лог.выр-е>

Другая особенность оператора REPEAT-UNTIL – в теле цикла (между ключевыми словами REPEAT и UNTIL) разрешается записать любое количество операторов, т.е. ключевые слова REPEAT и UNTIL сами выступают в роли операторных скобок.

Оператор цикла повторений ( FOR )

Цикл FOR (цикл повторений): выполняет заданное количество итераций; в выделенной переменной-счётчике сохраняется номер текущей итерации. Правила подсчёта итераций в Pascal: в начале цикла счётчику присваивается стартовое значение, после каждой итерации значение счётчика увеличивается (либо уменьшается) на единицу – и так до тех пор, пока значение счётчика не достигнет заданного конечного значения, – в этом случае выполняется последняя итерация, и цикл завершается. В Pascal есть две формы цикла FOR – (1) когда счётчик увеличивается и (2) когда счётчик уменьшается.

Цикл FOR с увеличением счётчика:


<for-оператор> ::= for <счётчик> := <Min.знач.> to <Max.знач.> do <оператор>

Цикл FOR с уменьшением счётчика:


<for-оператор> ::= for <счётчик> := <Max.знач.> downto <Min.знач.> do <оператор>

Счётчик – это имя переменной перечислимого типа, – целое число, символ, boolean (или к.-л. другой перечислимый тип).
ВАЖНО! Вещественные числа не являются перечислимым типом и не могут использоваться в качестве счётчика цикла FOR.
Синтаксическая диаграмма оператора WHILE:

Синтаксическая диаграмма оператора REPEAT-UNTIL:

Синтаксическая диаграмма оператора FOR:

Блок-схемы, экивалентные операторам циклов:

Примеры использования циклов:

Вычисление N!
	WHILE
	REPEAT-UNTIL
	FOR

	Var
  i,N,F: integer;

Begin

Readln(N);

F:=1; 
i:=2;

While i<=N do
  begin
    F:=F*i;

    i:=i+1;

  end;

Writeln(N,'! = ', F);

End;
	Var
  i,N,F: integer;

Begin

Readln(N);

F:=1;
i:=1;

Repeat

  F:=F*i;

  i:=i+1;

Until i>N do

Writeln(N,'! = ', F);

End;
	Var

  i,N,F: integer;

Begin
Readln(N);

F:=1;
for i:=2 to N do
  F:=F*i;

Writeln(N,'! = ', F);

End;


Суммирование нечётных чисел до N:

	WHILE
	REPEAT-UNTIL
	FOR

	Var

  i,N,F: integer;

Begin

Readln(N);

F:=0; 

i:=1;

While i<=N do
  begin
    F:=F+i;

    i:=i+2;
  end;

Writeln(N,'! = ', F);

End;
	Var

  i,N,F: integer;

Begin

Readln(N);

F:=0;

i:=1;

Repeat

  F:=F+i;

  i:=i+2;
Until i>N do

Writeln(N,'! = ', F);

End;
	Var

  i,N,F: integer;

Begin
Readln(N);

F:=0;

for i:=0 to ((N-1) div 2) do
  F:=F+(2*i+1);
Writeln(N,'! = ', F);

End;


Особенности цикла FOR

(1) Значения <Min.знач.> и <Max.знач.> могут быть любыми, совместимыми с типом счётчика цикла по присваиванию. Пример программы, распечатывающей символы и их ASCII-коды (только для заглавных английских букв):

Var

  c: char;

Begin
for c:= 'A' to 'Z' do
  Writeln(c, ' =', ord(c));

End;

Как добиться, чтобы в цикле FOR индекс менялся с произвольным шагом

В цикле FOR индекс всегда меняется с шагом +1 (либо –1). В Delphi запрещено изменять значение счётчика цикла внутри цикла! Это ограничение проверяется ещё на этапе компиляции программы, и связано с поддержкой оптимизации исполняемого кода. Добиться "произвольного шага" для счётчика цикла возможно только с помощью вычисления вспомогательной величины.
Пример – суммирование N первых нечётных чисел:

Var

  i, k, n, Sum: integer;

Begin

Readln(n);

Sum:=0;

for i:= 1 to n do begin

  k:=i*2-1;
// k – вспомогательная величина, i-е нечётное число
  Sum:=Sum+k;

  end;

End;
Если требуется более сложное изменение индекса цикла, то следует подумать о замене цикла FOR на цикл WHILE, который представляет более общую форму циклов.

Вложенные циклы

Внутри оператора цикла можно использовать любые другие операторы Паскаля, в том числе – операторы циклов. В таком случае операторы циклов будут называться вложенными. Пример: генерация 3-значных чисел, у которых сумма цифр равна 10:

Var
  a, b, c, N: integer;

Begin

for a:= 1 to 9 do
// 1-я цифра может быть любой от 1 до 9
  for b:= 0 to 10-a do
// 2-я цифра может иметь несколько значений,

    begin
//     в зависимости от значения 1-й цифры
    c:= 10-a-b;
// 3-я цифра однозначно вычисляется по двум первым
    N:= a*100 + b*10 +c;
// число формируется по значениям цифр
    Writeln(N);

    end;

End;
Простые и составные типы данных

Простые, или элементарные, типы данных: переменные этих типов считаются "неделимыми". К ним относятся: целочисленные (integer, cardinal, byte, word, shortint, smallint, longint, int64 и производные типы) и вещественные типы (real, single, double, extended), символы (char, widechar и производные типы), логический тип (boolean), указатели и перечисления. Составные типы данных – переменные этих типов имеют внутреннюю структуру – они состоят из нескольких переменных простых или составных типов, причём к этим внутренним переменным можно обращаться по отдельности. К составным типам относятся: массивы (array), строки (string), записи (record), файлы (file), множества (set), объекты (object) и классы (class).

Перечислимые типы

Перечислимые типы данных: для значений данного типа естественно определяются понятия "следующее значение" и "предыдущее значение". К ним относятся: все целые типы, все символьные типы, boolean, перечисления и производные от них интервальные типы. Вещественные типы данных не относятся к перечислимым, поскольку для вещественных чисел не определены понятия "предыдущее / следующее значение". Для любых перечислимых типов определены функции:

	Функция
	Описание

	Pred(X)
	Выдаёт значение, предшествующее X. Тип результата совпадает с типом X. Для целых чисел: Pred(X) ( X-1

	Succ(X)
	Выдаёт значение, следующее за X. Тип результата совпадает с типом X.       Для целых чисел: Succ(X) ( X+1

	Low(X)
	Выдаёт наименьшее значение перечислимого типа

	High(X)
	Выдаёт наибольшее значение перечислимого типа

	Ord(X)
	Порядковый номер величины X в соответсвующем типе данных. Для символов (тип char) Ord(X) = код символа.


	Процедура
	Описание

	Inc(X)
	Увеличивает значение переменной X
(аналог: X = Succ(X) )

	Dec(X)
	Уменьшает значение переменной X
(аналог: X = Pred(X) )

	Inc(X,n)
	n раз увеличивает значение переменной X 

	Dec(X,n)
	n раз уменьшает значение переменной X


Интервальные (отрезковые) типы

Интервальный (отрезковый) тип – новый тип данных, значения которого лежат в ограниченном диапазоне значений базового перечислимого типа, от заданного минимального до заданного максимального. Синтаксис объявления типа:

Пример

Type

   Day_Of_Month = 1..31;

В примере объявлен новый тип Day_Of_Month – отрезок целых чисел, от 1 до 31.

Массивы

Переменные‑массивы состоят из нескольких элементов – переменных одного типа (как простого, так и составного). Элементы различаются по индексу элемента в массиве. Индекс(ы) массива должен быть значением какого-нибудь перечислимого типа (целое число, символ, boolean, перечисление, отрезковый тип). Допускается объявление как переменной-массива с определённой структурой, так и нового типа данных – массивов с определённой структурой.

С элементами массива работают так же, как с обычными переменными. При использовании элемента массива надо указать только имя переменной‑массива и индекс(ы) элемента. В Pascal и Delphi допускаются одномерные массивы (элементы имеют один индекс) и многомерные массивы (элементы определяются несколькими индексами). Макс. размерность массива = 256 (до 256 индексов). 

Синтаксис объявления типа массива:

Примеры объявлений массивов:

Var

   Days_Per_Month:  array[1..12] of 1..31;            // 1-мерный массив из 12 элементов
   DayOfWeek_Names: array[1..7]  of string;                // 1-мерный массив строк
Type

   TClassSchedule = array[1..6, 1..8] of string;           // 2-мерный массив строк
   Type1_3D_Array = array['0'..'9',-109..89,byte] of real; // 3-мерный массив вещ.чисел
   Type2_3D_Array = array[char,char,char] of string;       // 3-мерный массив строк
Var

   SchoolSchedule:  array[10..11, 'А'..'Ж'] of TClassSchedule; // массив массивов
Примеры использования массивов:

Days_Per_Month[1] := 31;                              // обращение к элементам массива
Days_Per_Month[2] := 28;

Days_Per_Month[3] := Days_Per_Month[1];

DayOfWeek_Names[1]:= 'Понедельник';

DayOfWeek_Names[7]:= 'Воскресенье';

SchoolSchedule[11,'Е', 2,1]  := 'Информатика';               // примеры разного указания
SchoolSchedule[11,'Е'][2,1]  := SchoolSchedule[11,'Е', 2,2]; // индексов массива
SchoolSchedule[11]['Ж'][2][3]:= 'Информатика'; 

Логический тип данных (BOOLEAN):

В Паскале/Delphi имеется особый тип данных для логических значений (истина/ложь), он называется логическим типом данных, имя этого типа (в программах) – boolean. Переменные типа boolean могут иметь только 2 значения – FALSE (ложь), или TRUE (истина). FALSE и TRUE – это логические константы (их имена также являются ключевыми словами языка). Процедуры вывода тескта на консоль (и в текстовый файл) Read и Readln печатают значения логических переменных и выражений как текст FALSE и TRUE, соответственно.

Тип boolean является перечислимым: Ord(FALSE)=0; Ord(TRUE)=1. А поэтому: можно объявлять массивы с индексами типа boolean, а также можно организовывать циклы for с индексом типа boolean. Это свойство можно использовать для программирования таблиц истинности логических операций и функций.

Пример программирования таблицы истинности для логической операции импликации

Var

   A, B, temp, Conclusion: boolean;

   Imp: array [boolean, boolean] of boolean;  // таблица истинности импликации:
   X, Y, h: real;                             //   Imp[A,B] эквивалентно значению A=>B
Begin                               // Определяем значения для импликации 

                                    // для всех вариантов аргументов:

Imp[FALSE, FALSE] := TRUE;          //    ложь => ложь   =  истина
Imp[FALSE, TRUE ] := TRUE;          //    ложь =>истина  =  истина
Imp[TRUE , FALSE] := FALSE;         //   истина=> ложь   =  ложь
Imp[TRUE , TRUE ] := TRUE;          //   истина=>истина  =  истина
                 // Пример использования таблицы истинности

                 // проверяем, правда ли, что если X>Y, то sin(X)>sin(Y)

h := pi/100;                        //   h = шаг изменения X и Y
X := -pi;                           // будем проверять значения X и Y в интервале от -( до +(
Y := -pi;

Conclusion := TRUE;                 // Conclusion = флаг истинности гипотезы
Repeat 
   Repeat 

      temp := Imp[ (X>Y), (sin(X)>sin(Y)) ];     // вычисляем истинность выражения для
                                                 //  конкретных значений X и Y
      Writeln('X=',X, ' Y=',Y, ' is sin(X)>sin(Y) when X>Y : ',temp);  // печатаем 

      Conclusion := Conclusion and temp;         // добавляем текущий результат к общему 

      Y := Y+h;                     // меняем значение Y
   Until (Y>pi);

   X := X+h;                        // меняем значение X
Until (X>pi);

if Conclusion 
then  Writeln('Our hypothesis is valid')

else  Writeln('Our hypothesis is wrong!');

End;

Логические операции:

	Логическая операция
	Знак операции в Паскале/Delphi
	Примеры

	отрицание
	(A, ( A
	not
	B := not A;

	конъюнкция
	A&B, A(B
	and
	C := A and B;

	дизъюнкция
	A(B
	or
	C := A or B;

	строгая дизъюнкция
	A(B
	xor
можно заменить на <>
	C := A xor B;
C := A <> B;

	эквивалентность
	A = B
	=
	C := A = B;

	импликация
	A(B
	можно заменить на <=
	C := A <= B;


Приоритеты логических операций:

1)       not 
2)       and 

3)    or  и  xor 

4) операции сравнения   =,   >,   <,   <>,   <=,   >=

Битовые операции:

Для целых чисел определены побитовые операции – они меняют значения битов числа (интерпретируя каждый бит как логическое значение: 0 = ложь, 1 = истина).

	not
	инверсия всех битов
	B := not A
	A = 0010 0110   ==>  B = 1101 1001

	and
	логическое умножение всех битов
	C := A and B
	A = 0010 0111, B = 1000 1111 ==> C = 0000 0111

	or
	логическое сложение всех битов
	C := A or B
	A = 0010 0111, B = 1000 1111 ==> C = 1010 1111

	xor
	сложение по модулю 2 всех битов
	C := A xor B
	A = 0010 0111, B = 1000 1111 ==> C = 1010 1000

	shl
	сдвиг всех битов "влево"
(в сторону старших разрядов)
	B := A shl 2
	A = 0010 0111 ==> B = 1001 1100

	shr
	сдвиг всех битов "вправо"
(в сторону младших разрядов)
	B := A shr 1
	A = 0010 0111 ==> B = 0001 0011


Работа со строками в Delphi
Строки (string) – это один из составных типов данных, используются для обработки текстовых данных. Строка – это текст, размещённый в памяти компьютера. Элементами строки являются символы (тип char), они индексируются (нумеруются), начиная с индекса 1. Обращение к символам-элементам строки аналогично обращению к элементам массива. Примеры:

Var
  S: string;
// объявление строковой переменной
  c: char;
// символьная переменная
  n: integer; 
// целочисленная переменная
Begin
s := 'hello world';
// присвоение строкового значения строковой переменной
c := s[5];
// 5-й символ строки сохраняется в символьной переменной, c='o'
s[1] := 'H';
// 1-й символ строки меняется на 'H' (s='Hello world')
n := Length(s);
// получение длины строки
End;

Задание строковых констант (выражений)

1) Нужный текст заключается в одинарные кавычки:
'important message'
2) Если текст должен содержать одинарные кавычки, то каждую одинарную кавычку в тексте надо заменить на ДВА подряд идущих символа одинарной кавычки:
'It''s an example text containing single quote symbols ('') '
Также в строку можно включать символы с любыми кодами, в т.ч. такие, которые невозможно набрать на клавиатуре:

3) любой символ можно добавить к строке, используя его код (номер):
#0, #9, #13, #65, #128, #255.

4) символы с кодами от #1 до #26 допускается обозначать так:
^A, ^B, ^C,...^Z.

Символы, заданные кодами, необходимо записывать перед/после/между фрагментами строк, заключённых в одинарные кавычки.

Примеры корректных строковых констант:

'First line'#13'Second line'#13'Third line'^M'Fourth line'^'Fifth line'

#1#2#3#4#5#6'Ку-ку-кубик'#9'Ква-ква-квадратик'

'^M - new line'^M'^H - beep'^H^M'^J - tab'^J^J'...'

Операции со строками:

В Delphi и Паскале для строк допустимы следующие операции:

	Обращение к символу строки по номеру символа
	s[1] - первый симовл строки, s[k] - k-й символ строки

	Объединение строк (конкатенация), строки и символа
	s1+s2,   s+'!',   'Hello'+' '+name

	Сравнение значений строк
	S1=S2, S1>S2, S1<S2, S1>=S2, S1<=S2, S1<>S2


Строки сравниваются в т.н. лексикографическом порядке: строки сравниваются посимвольно, начиная с 1-х символов. Если первые символы совпадают, то переходят к сравнению вторых символов и т.д., пока не будет найдена пара несовпадающих символов, либо одна из строк не закончится. Если кончилась одна из строк, более короткая считается меньшей, если найдены несовпадающие символы, меньшей считается та строка, у которой на данной позиции стоит символ с меньшим кодом. Примеры:

	'ABC' < 'ABCD'
	'ABC' > 'ABBяяяяяяяяя'
	'1234556' < '2'
	'Z' < 'a'

	#255 > 'ZZZZZZZZZZ'
	^M'   ' < 'M'
	'ABC' < 'abc'
	#68#100 = 'Dd'


Пустая строка - строка нулевой длины, задаётся так:
S := '';
Строковые функции и процедуры:

	Функция
	Описание

	Length(S)
	Выдаёт длину строки S (сколько символов в строке)

	Copy(S, k, n)
	Возвращает подстроку (часть строки) из n символов, начиная с k-го символа строки S.

	UpperCase(S)
	Возвращает копию исходной строки S, у которой все английские буквы переведены в верхний регистр

	AnsiUpperCase(S)
	Возвращает копию исходной строки S, все английские и русские буквы переведены в верхний регистр

	LowerCase(S)
	Возвращает копию исходной строки S, у которой все английские буквы переведены в нижний регистр

	AnsiLowerCase(S)
	Возвращает копию исходной строки S, все английские и русские переведены в нижний регистр

	Pos(substr, S)
	Ищет подстроку substr в строке S, если нашла - возвращает номер первого символа подстроки, иначе - 0.

	Trim(S)
	возвращает копию строки S, у которой удалены пробелы и табуляции перед после текста.

	SameText(S1, S2)
	Сравнивает две строки, игнорируя различия в регистре английских букв

	AnsiSameText(s1,s2)
	Сравнивает две строки, игнорируя различия в регистре английских и русских букв


	Процедура
	Описание

	Write(S)
	Выводит на экран содержимое строки S. спец. символы интерпретируются как управляющие.

	Readln(S)
	Считывает строку с клавиатуры. В строку попадает весь текст строки, включая пробелы и т.д.

	Insert(substr,S,n)
	Вставляет подстроку substr в строку S так, что 1-й символ substr становится n-м символом строки S.

	Delete(S, k, n)
	Удаляет символы из строки S: n символов, начиная с k-го символа.

	Str(X[:n[:m]],S)
	В строку S записывает текстовое представление значения X, так, как это делают Write и Writeln

	Val(S, X, j)
	Пытается прочесть число X из строки S. В целую переменную j помещается номер символа, который не может быть частью записи числа, либо 0, если строка содержала запись ровно одного числа.


Var
   z: real = 1234.56789;
// объявили веществ. переменную и задали начальное значение
   s: string;

   k, n: integer;

Begin
Str(z:0:5, s);

// Получим s = '1234.56789'
Val(s, n, k);

// Получим n=1234 и k=5
End; 


//   (т.к. символ '.' не может присутствовать в записи целого числа)
Работа с текстовыми файлами

Файлы (files) – это способ хранения данных на устройствах длительного хранения информации – жёстких дисках, CD/DVD, FLASH, сетевых ресурсах и т.д. Файлы размещаются в файловых системах. Для файловых систем, файл – это единица хранения информации. Файл имеет такие характеристики, как имя, расширение, путь, размер, атрибуты, права доступа и др. Полное имя файла (fully-qualified file name) состоит из имени устройства (буквы диска или имени сервера), пути (списка каталогов), собственно имени файла и расширения (MIME-расширения): C:\Windows\System32\config\SYSTEM.SAV
Текстовые файлы

Любые файлы хранят просто последовательность байтов. Программы, читающие данные из файла, интерпретируют их содержимое как некий документ определённого формата (TXT, DOC, AVI, MP3, EXE и т.п.). Текстовые файлы содержат только текст – буквы, цифры, знаки препинания и др.символы, которые обычно встречаются в печатных текстах, например, в книгах.

В Дельфи текстовые файлы читаются через функции-интерпретаторы консольного ввода-вывода. Поэтому чтение-запись для текстовых файлов во многом похожа на работу с консолью. Как на деле устроен консольный ввод-вывод: в операционной системе файлы с пустым именем ‘’ и именем 'CON' (в Windows) ассоциирован с консолью. Если такой файл открыть на чтение, из файла читается то, что  печатается в консоли программы. Если открыть на запись – запись в файл выводится на экран. В Паскале/Дельфи есть две спец.файловых переменных INPUT и OUTPUT, связанные с файлом ‘’, они определены в любой консольной программе.

Программная работа с текстовыми файлами.

Тип переменных для текстовых файлов: text или textfile.

Функции:

	AssignFile(F, filename)
	Задать имя файла – связать файловую переменную с реальным файлом. 

	Reset(F)
	Открыть файл для чтения. Запись в файл невозможна, пока его не откроют в другом режиме.

	Rewrite(F)
	Открыть файл для записи. Чтение из файла невозможно, пока его не откроют в режиме Reset. Исходное содержимое файла уничтожается при открытии файла, его длина обнуляется.

	Append(F)
	Открыть файл для до-записи – курсор помещается в самый конец файла. Исходное содержимое файла НЕ удаляется. Запись новых данных происходит в конец файла.

	Read, ReadLn
	Чтение данных из текстового файла. Синтаксис вызова – как для консольного ввода.


	Write, WriteLn
	Запись данных в текстовый файл. Синтаксис вызова – как для консольного вывода.

	EOF(F)
	Проверка на ситуацию "достигнут конец файла". Возвращает значение типа BOOLEAN.

	SeekEOF(F)
	Как EOF, только заодно пропускает "пустой текст" - пробелы, табуляции и пустые строки.

	EOLn(F)
	Проверка на ситуацию "достигнут конец строки". Возвращает значение типа BOOLEAN.

	SeekEOLn(F)
	Как EOLn, только заодно пропускает "пустой текст" - пробелы и табуляции.

	SetTextBuf(F,buff,buffsize)
	Позволяет задать большой буфер для текстового файла, что значительно ускоряет ввод-вывод.


Функции управления файлами в файловых системах:

	FileSize(F)
	Вернуть длину файла в байтах
	DirectoryExists(dirname)
	Проверяет наличие папки

	RenameFile(oldname,newname)
	Переименовать файл
	CreateDir(dirname)
	Создать папку (1 уровень)

	DeleteFile(filename)
	Удалить файл
	ForceDirectories(path)
	Создать путь (≥1 уровня)

	FileExists(filename)
	Проверяет наличие файла
	GetDir
	Вурнуть текущий путь


Примеры работы с текстовыми файлами.

	Var

// Подсчёт строк в файле


T: text;
// файловая переменная

S: string;


i,j: integer;


Buf: string;

BEGIN


//-задать буфер для файла-

SetLength(Buf, 1000000);  // задать длину буфера
SetTextBuf(T, Buf[1], Length(Buf));

AssignFile(T, ‘myfile.txt’);

Reset(T);
// открыть на чтение
i:=0;

While not EOF(F) do begin // пока не конец файла

Readln(T,S);
// чтение строк

Inc(i);


if (S>’’)


then Inc(J);


end;

CloseFile(T);
// закрыть файл


//--печать результатов--

Writeln(‘В этом файле строк:');

Writeln(#9‘всего   =’, i);

Writeln(#9‘непустых=’, j);

Writeln(#9‘пустых  =’, i-j);

END.
	Var

//-- Запись массива в файл


T: text;
// файловая переменная

R: array of real;


i: integer;

BEGIN

SetLength(R, 1+Random(1000000));

for i:=0 to Length(R)-1 do

R[i]:=i*10+Random(100);

AssignFile(T, ‘array.txt’);

Rewrite(T); 
// открыть на запись
for i:=0 to Length(R)-1 do begin

Write(T,R[i]:20, '  ');
// запись чисел

if (i+1) mod 3 =0 
// по 3 на строке

then Write(T);


end;

CloseFile(T);

// закрыть файл
END.

	
	Var

//- Чтение массива из файла


T: text;
// файловая переменная

R: array of real;


n: integer;

BEGIN

SetLength(R, 1000000);  // задать длину массива
AssignFile(T, ‘array.txt’);

Reset(T);
// открыть на чтение

n:=0;
While not SeekEOF(F) do begin

Read(T,R[i]);
// прочитать число

Inc(n); 
// увеличить счётчик

end;

SetLength(R, n);
// исправить длину массива
CloseFile(T);
// закрыть файл

Writeln('Прочитано чисел: ', n);

END.


Процедуры и функции (подпрограммы)

Подпрограммы: в больших программных проектах всегда имеется много мелких подзадач, которые приходится неоднократно выполнять во время работы большой программы, например: вводить данные с клавиатуры, рисовать графики и диаграммы, искать файл на диске или данные в базе данных, или вычислять интеграл функции на разных отрезках, и т.п. Чтобы не писать в программе множество копий одного и того же алгоритма, его оформляют в виде подпрограммы (у которой имеется собственное имя), а затем вызывают подпрограмму по имени. Примеры – все математические и прочие стандартные функции Паскаля: SIN, COS, ARCTAN, RANDOM, Round, Trunc, Int, Frac, Readln, Writeln, IntToStr, DateTimeToStr и т.д.

В Паскале/Дельфи имеются 2 вида подпрограмм – процедуры и функции. Функции обязательно возвращают какое-либо значение как результат работы функции (например, sin, Round, Random), процедуры – делают что-либо полезное, но не возвращают обратно значения как результат своей работы. Примеры: Writeln – печатает значения чисел на экран, но ничего не возвращает в программу в качестве результата своей работы; Readln – считывает значения переменных с клавиатуры, т.е. меняет их, но ничего не сообщает в программу ​– удалось прочитать или нет, сколько символов пришлось прочесть, или сколько времени ждать, – НИЧЕГО.

Обычно подпрограммам необходимо передать т.н. аргументы – входные данные, с которыми они будут работать. Внутри себя подпрограммы могут дополнительно использовать внутренние переменные.

